01006 Advanced Engineering Mathematics 1
2-hr test May 11 2023

JE 9.5.23 (translated by SHSP)

Problem 1

```
> restart;with(LinearAlgebra):with(plots):
```

A function f: $\mathbb{R}^2 \to \mathbb{R}$ is given by the expression:

```
> f:=(x,y)->x^2*y-3*x^2-4*y^2+8*y+11:
> f(x,y)
```

$$x^2 y - 3x^2 - 4y^2 + 8y + 11 \quad \textbf{(1.1)}$$

Question 1

$\nabla f(x, y) = (f'_x(x, y), f'_y(x, y)) = (2xy - 6x,\ x^2 - 8y + 8) = (2x(y-3), x^2 - 8y + 8) = (0, 0)$
$2x(y-3) = 0$ and $x^2 - 8y + 8 = 0 \Leftrightarrow x = 0$ and $y = 1$ or $y = 3$ and $x = \pm 4$.
All stationary points of f are then (0, 1), (4, 3) and (−4, 3).

Question 2

If f has a local extremum at a point then that point must be a stationary point since f has no exceptional points.
The Hessian matrix of f at the points (x, y) is

```
> H(x,y):=<diff(f(x,y),x,x),diff(f(x,y),x,y);diff(f(x,y),y,x),
  diff(f(x,y),y,y)>
```

$$H(x, y) := \begin{bmatrix} 2y - 6 & 2x \\ 2x & -8 \end{bmatrix} \quad \textbf{(1.2.1)}$$

```
> H(0,1):=subs(x=0,y=1,H(x,y))
```

$$H(0, 1) := \begin{bmatrix} -4 & 0 \\ 0 & -8 \end{bmatrix} \quad \textbf{(1.2.2)}$$

```
> Eigenvalues(H(0,1),output=list)
```

$$[-4, -8] \quad \textbf{(1.2.3)}$$

Since both eigenvalues of $\mathbf{H}(0, 1)$ are negative, then f has a proper local maximum at the stationary point (0, 1).

```
> H(4,3):=subs(x=4,y=3,H(x,y))
```

$$H(4, 3) := \begin{bmatrix} 0 & 8 \\ 8 & -8 \end{bmatrix} \quad \textbf{(1.2.4)}$$

```
> Eigenvalues(H(4,3),output=list)
```

(1.2.5)

$$\left[-4+4\sqrt{5},\ -4-4\sqrt{5}\right] \quad \textbf{(1.2.5)}$$

Since the two eigenvalues of **H**(4, 3) have opposite signs, then f has neither local maximum nor local minimum at the stationary point (4, 3) (saddle point).

```
> H(-4,3):=subs(x=-4,y=3,H(x,y))
```

$$H(-4,3) := \begin{bmatrix} 0 & -8 \\ -8 & -8 \end{bmatrix} \quad \textbf{(1.2.6)}$$

```
> Eigenvalues(H(-4,3),output=list)
```

$$\left[-4+4\sqrt{5},\ -4-4\sqrt{5}\right] \quad \textbf{(1.2.7)}$$

Since the two eigenvalues of **H**(−4, 3) have opposite signs, then f has neither local maximum nor local minimum at the stationary point (−4, 3) (saddle point).

```
> contourplot(f(x,y),x=-6..6,y=-2..5,contours=40)
```



Question 3

A parabola with the equation $y = \frac{1}{4}x^2 - 1$ along with a straight line with the equation $y = 3$ delimit a bounded and closed set of points M.

```
> implicitplot({y=1/4*x^2-1,y=3},x=-4..4,y=-1..3,scaling=
```

```
constrained)
```



Since M is bounded and closed and since f is continuous on M, then f has a global minimum and a global maximum on M. Since f does not have any exceptional points in the interior of M, then these values are found either at the stationary point or within the interior of M or on the boundary of M.

The only stationary point in the interior of M is (0, 1) where we have the function value

```
> 'f(0,1)'=f(0,1)
```

$$f(0, 1) = 15 \quad \textbf{(1.3.1)}$$

The value of f on the two boundary curves is

```
> simplify(f(x,1/4*x^2-1))
```

$$-1 \quad \textbf{(1.3.2)}$$

for all $x \in$ [-4; 4] and

```
> f(x,3)
```

$$-1 \quad \textbf{(1.3.3)}$$

for all $x \in$ [-4; 4].

From numerical comparison of these investigations we find that the global maximum is 15, which is achieved at the point (0, 1), and that the global minimum is −1, which is achieved everywhere on the boundary of M.

We note that since M is connected, then the image set (the range) is $f(M)$ = [−1; 15].

```
> plot3d(f(x,y),x=-4..4,y=-1..3)
```



Problem 2

```
> restart;with(LinearAlgebra):with(plots):
```

A region M in the (x, z) plane is given by

$M = \{(x, z) \mid 0 \quad x \le \frac{1}{4}\pi \text{ and } \sin(x) \le z \le \cos(x)\}$

```
> p0:=plot([cos(x),sin(x)],x=0..Pi/4,scaling=constrained,color=
  black):
> p1:=plot([Pi/4,v,v=0..1],linestyle=dash):
```

```
> display(p0,p1,tickmarks=[2,3],labels=[x,z],title="M")
```



Question 1

A parametric representation of M is

```
> s:=(u,v)-><u,sin(u)+v*(cos(u)-sin(u))>:
> s(u,v)
```

$$\begin{bmatrix} u \\ \sin(u) + v\,(\cos(u) - \sin(u)) \end{bmatrix} \quad \textbf{(2.1.1)}$$

where $u \in [0;\ \frac{\pi}{4}]$ and $v \in [0;\ 1]$ (the vertical line segment from the point (u,sin(u)) to the point (u, cos(u))).

```
> M:=<diff(s(u,v),u)|diff(s(u,v),v)>
```

$$M := \begin{bmatrix} 1 & 0 \\ \cos(u) + v\,(-\sin(u) - \cos(u)) & \cos(u) - \sin(u) \end{bmatrix} \quad \textbf{(2.1.2)}$$

```
> Js:=Determinant(M)
```

$$Js := \cos(u) - \sin(u) \quad \textbf{(2.1.3)}$$

which is ≥ 0, since $u \in [0; \frac{\pi}{4}]$. The Jacobian function belonging to **s** is thus

```
> Jacobi:=Js
```

$$Jacobi := \cos(u) - \sin(u) \tag{2.1.4}$$

$$\mathrm{Ar}(M) = \int_M \mathrm{d}\mu = \int_0^1 \int_0^{\frac{\pi}{4}} \mathrm{Jacobian}(u,v)\mathrm{d}u\mathrm{d}v$$

```
> integranden:=Jacobi
```

$$integranden := \cos(u) - \sin(u) \tag{2.1.5}$$

```
> Int(integranden,[u=0..Pi/4,v=0..1])=int(integranden,[u=0..
  Pi/4,v=0..1])
```

$$\int_0^1 \int_0^{\frac{\pi}{4}} (\cos(u) - \sin(u))\, \mathrm{d}u\, \mathrm{d}v = \sqrt{2} - 1 \tag{2.1.6}$$

A solid of revolution Ω has the parametric representation

```
> r:=(u,v,w)-><u*cos(w),u*sin(w),sin(u)+v*(cos(u)-sin(u))>
```

$$r := (u, v, w) \mapsto \langle u \cdot \cos(w), u \cdot \sin(w), \sin(u) + v \cdot (\cos(u) - \sin(u)) \rangle \tag{2.1}$$

```
> r(u,v,w)
```

$$\begin{bmatrix} u \cos(w) \\ u \sin(w) \\ \sin(u) + v\, (\cos(u) - \sin(u)) \end{bmatrix} \tag{2.2}$$

where $u \in [0; \frac{\pi}{4}]$, $v \in [0; 1]$ and $w \in [0; \pi]$.

We note that Ω has been created by rotation of region *M* in the (*x*, *z*) plane by an angle of π about the *z* axis counter-clockwise as seen from the positive end of the *z* axis.

Question 2

```
> M:=<diff(r(u,v,w),u)|diff(r(u,v,w),v)|diff(r(u,v,w),w)>
```

$$M := \begin{bmatrix} \cos(w) & 0 & -u \sin(w) \\ \sin(w) & 0 & u \cos(w) \\ \cos(u) + v\, (-\sin(u) - \cos(u)) & \cos(u) - \sin(u) & 0 \end{bmatrix} \tag{2.2.1}$$

```
> Jr:=simplify(Determinant(M))
```

$$Jr := -u\, (\cos(u) - \sin(u)) \tag{2.2.2}$$

which is ≤ 0, since $u \in [0; \frac{\pi}{4}]$. The Jacobian function belonging to **r** is thus

```
> Jacobi:=-Jr
```

$$Jacobi := u\, (\cos(u) - \sin(u)) \tag{2.2.3}$$

$$\mathrm{Vol}(\Omega) = \int_{\Omega} \mathrm{d}\mu = \int_0^{\pi}\int_0^1\int_0^{\frac{\pi}{4}} \mathrm{Jacobian}(u,v,w)\mathrm{d}u\mathrm{d}v\mathrm{d}w$$

```
> integranden:=Jacobi
```

$$integranden := u\,(\cos(u) - \sin(u)) \qquad \textbf{(2.2.4)}$$

```
> Int(integranden,[u=0..Pi/4,v=0..1,w=0..Pi])=int(integranden,
  [u=0..Pi/4,v=0..1,w=0..Pi])
```

$$\int_0^{\pi}\int_0^1\int_0^{\frac{\pi}{4}} u\,(\cos(u) - \sin(u))\,\mathrm{d}u\,\mathrm{d}v\,\mathrm{d}w = \left(-1 + \frac{\pi\sqrt{2}}{4}\right)\pi \qquad \textbf{(2.2.5)}$$

Problem 3

```
> restart;with(LinearAlgebra):with(plots):
> vop:=proc(X) op(convert(X,list)) end proc:
```

In the (x, y) plane we are given the velocity vector field $\mathbf{V}(x, y) = \begin{bmatrix} \frac{1}{2}x - \frac{1}{2}y \\ -\frac{1}{2}x + \frac{1}{2}y \end{bmatrix}$.

Question 1

For determination of the flow curves of **V**, we have the differential equation system

$$\begin{bmatrix} x'(t) \\ y'(t) \end{bmatrix} = \begin{bmatrix} \frac{1}{2}x(t) - \frac{1}{2}y(t) \\ -\frac{1}{2}x(t) + \frac{1}{2}y(t) \end{bmatrix} = \begin{bmatrix} \frac{1}{2} & -\frac{1}{2} \\ -\frac{1}{2} & \frac{1}{2} \end{bmatrix} \begin{bmatrix} x(t) \\ y(t) \end{bmatrix}, \; t \in \mathbb{R}.$$

```
> A:=<1/2,-1/2;-1/2,1/2>
```

$$A := \begin{bmatrix} \frac{1}{2} & -\frac{1}{2} \\ -\frac{1}{2} & \frac{1}{2} \end{bmatrix} \qquad \textbf{(3.1.1)}$$

```
> Eigenvectors(A,output=list)
```

$$\left[\left[1, 1, \left\{\begin{bmatrix} -1 \\ 1 \end{bmatrix}\right\}\right], \left[0, 1, \left\{\begin{bmatrix} 1 \\ 1 \end{bmatrix}\right\}\right]\right] \qquad \textbf{(3.1.2)}$$

All flow curves of **V** are thus given by

$$\begin{bmatrix} x(t) \\ y(t) \end{bmatrix} = c_1 \begin{bmatrix} 1 \\ 1 \end{bmatrix} + c_2 \begin{bmatrix} -1 \\ 1 \end{bmatrix} e^t = \begin{bmatrix} c_1 - c_2 e^t \\ c_1 + c_2 e^t \end{bmatrix}, t \in \mathbb{R}, c_1, c_2 \in \mathbb{R}.$$

For determination of the constants c_1 and c_2, then $\begin{bmatrix} x(0) \\ y(0) \end{bmatrix} = \begin{bmatrix} 2 \\ 4 \end{bmatrix}$ and we have the linear equation system

$\begin{bmatrix} x(0) \\ y(0) \end{bmatrix} = \begin{bmatrix} c_1 - c_2 \\ c_1 + c_2 \end{bmatrix} = \begin{bmatrix} 2 \\ 4 \end{bmatrix}$. From this we find $c_1 = 3$ and $c_2 = 1$.

The wanted flow curve is thus

```
> r:=t-><3-exp(t),3+exp(t)>:
> r(t)
```

$$\begin{bmatrix} 3 - e^t \\ 3 + e^t \end{bmatrix} \quad \textbf{(3.1.3)}$$

where $t \in \mathbb{R}$.

The curve thus initiates at the point

```
> r(0)
```

$$\begin{bmatrix} 2 \\ 4 \end{bmatrix} \quad \textbf{(3.1.4)}$$

and continues through the point

```
> r(ln(3))
```

$$\begin{bmatrix} 0 \\ 6 \end{bmatrix} \quad \textbf{(3.1.5)}$$

A curve K has the parametric representation $\mathbf{s}(u) = \begin{bmatrix} u \\ u^2 \end{bmatrix}$, $u \in [0; 2]$.

Question 2

For determination of the constants c_1 and c_2 so $\begin{bmatrix} x(0) \\ y(0) \end{bmatrix} = \begin{bmatrix} u \\ u^2 \end{bmatrix}$, we have the linear equation system

$\begin{bmatrix} x(0) \\ y(0) \end{bmatrix} = \begin{bmatrix} c_1 - c_2 \\ c_1 + c_2 \end{bmatrix} = \begin{bmatrix} 1 & -1 \\ 1 & 1 \end{bmatrix} \begin{bmatrix} c_1 \\ c_2 \end{bmatrix} = \begin{bmatrix} u \\ u^2 \end{bmatrix}$ with the corresponding augmented matrix

```
> T:=<1,-1,u;1,1,u^2>
```

$$T := \begin{bmatrix} 1 & -1 & u \\ 1 & 1 & u^2 \end{bmatrix} \quad \textbf{(3.2.1)}$$

```
> c:=LinearSolve(T)
```

$$c := \begin{bmatrix} \frac{1}{2}u^2 + \frac{1}{2}u \\ \frac{1}{2}u^2 - \frac{1}{2}u \end{bmatrix} \quad \textbf{(3.2.2)}$$

The flow curve $\mathbf{r}(u, t)$ that fulfills $\mathbf{r}(u,0) = \begin{bmatrix} u \\ u^2 \end{bmatrix}$ thus has the parametric representation

```
> r:=(u,t)-><c[1]-c[2]*exp(t),c[1]+c[2]*exp(t)>:
> r(u,t)
```

$$\begin{bmatrix} \frac{u^2}{2} + \frac{u}{2} - \left(\frac{1}{2}u^2 - \frac{1}{2}u\right)\mathrm{e}^t \\ \frac{u^2}{2} + \frac{u}{2} + \left(\frac{1}{2}u^2 - \frac{1}{2}u\right)\mathrm{e}^t \end{bmatrix} \quad \textbf{(3.2.3)}$$

For $t = \ln(3)$ and $u \in [0; 2]$ we get

```
> r(u,ln(3))
```

$$\begin{bmatrix} -u^2 + 2u \\ 2u^2 - u \end{bmatrix} \quad \textbf{(3.2.4)}$$

which is a parametric representation of the curve that K has been transformed into at time $t = \ln(3)$
.

```
> p1:=plot([vop(r(u,ln(3))),u=0..2],scaling=constrained,color=
  red):
> p2:=plot([u,u^2,u=0..2],scaling=constrained,color=blue):
> p3:=pointplot([[2,4],[0,6]],symbol=solidcircle,color=[blue,
  red]):
> p4:=plot([3-exp(t),3+exp(t),t=0..ln(3)],linestyle=dot,color=
  darkgrey):
> display(p1,p2,p3,p4,labels=[x,y])
```



This question could of course also have been solved by use of dsolve.

Problem 4

```
> restart:with(plots):
  prik:=(x,y)->VectorCalculus[DotProduct](x,y):
  kryds:=(x,y)->convert(VectorCalculus[CrossProduct](x,y),Vector):
  vop:=proc(X) op(convert(X,list)) end proc:
  grad:=X->convert(Student[VectorCalculus][Del](X),Vector):
  div:=V->VectorCalculus[Divergence](V):
  rot:=proc(X) uses VectorCalculus;BasisFormat(false);Curl(X)   end
  proc:
> with(LinearAlgebra):
```

A vector field **V** in (x, y, z) space is given by

```
> V:=(x,y,z)-><-y*x,y^2+5,-y*z+5*z>:
> V(x,y,z)
```

$$\begin{bmatrix} -y\,x \\ y^2+5 \\ -y\,z+5\,z \end{bmatrix} \tag{4.1}$$

Regarding the given massive prism P with the corners O, A, B, C, E and F we refer to the figure in the problem sheet.

Question 1

```
> divV:=div(V)(x,y,z)
```

$$divV := 5 \tag{4.1.1}$$

```
> rotV:=unapply(rot(V)(x,y,z),[x,y,z]):
> rotV(x,y,z)
```

$$\begin{bmatrix} -z \\ 0 \\ x \end{bmatrix} \tag{4.1.2}$$

$\mathrm{Vol}(P) = \frac{1}{2}\,\mathrm{Vol}(\mathrm{box}) = \frac{4}{2} = 2.$

Question 2

∂P is the closed surface of P with an orientation given by an outwards-directed unit normal vector. From Gauss' Theorem we then get

$\mathrm{Flux}(\mathbf{V},\partial P) = \int_P \mathrm{Div}(\mathbf{V})\,\mathrm{d}\mu = \int_P 5\,\mathrm{d}\mu = 5\int_P \mathrm{d}\mu = 5\mathrm{Vol}(P) = 10.$

Let S denote the side surface of P that has the corners B, C, E and F.

Question 3

Since S is the rectangle that has the corners B, C, E and F located within a plane with the equation $z = 2 - y$, then a parametric representation of S is

```
> r:=(u,v)-><u,v,2-v>:
> r(u,v)
```

$$\begin{bmatrix} u \\ v \\ 2 - v \end{bmatrix} \tag{4.3.1}$$

where $u \in [0; 1]$ and $v \in [0; 2]$.

```
> ru:=diff(r(u,v),u)
```

$$ru := \begin{bmatrix} 1 \\ 0 \\ 0 \end{bmatrix} \tag{4.3.2}$$

```
> rv:=diff(r(u,v),v)
```

$$rv := \begin{bmatrix} 0 \\ 1 \\ -1 \end{bmatrix} \quad \textbf{(4.3.3)}$$

The normal vector of the surface

```
> N:=kryds(ru,rv)
```

$$N := \begin{bmatrix} 0 \\ 1 \\ 1 \end{bmatrix} \quad \textbf{(4.3.4)}$$

perfectly fulfills the right-hand rule with the chosen orientation of the closed boundary curve ∂S of S (show with red arrow on the figure in the problem sheet). From Stokes' Theorem we thus get

$$\mathrm{Circ}(\mathbf{V},\partial S) = \int_{\partial S} \mathbf{V} \cdot \mathbf{e}_{\partial S}\, \mathrm{d}\mu = \mathrm{Flux}(\mathbf{Curl}(\mathbf{V}),S) = \int_S \mathbf{n}_S \cdot \mathbf{Curl}(\mathbf{V})\mathrm{d}\mu = \int_0^2 \int_0^1 \mathbf{N}(u,v) \cdot \mathbf{Curl}(\mathbf{V})(\mathbf{r}(u.v))\, \mathrm{d}u\mathrm{d}v$$

The curl computed on the surface

```
> Rot:=rotV(vop(r(u,v)))
```

$$Rot := \begin{bmatrix} -2+v \\ 0 \\ u \end{bmatrix} \quad \textbf{(4.3.5)}$$

```
> integranden:=prik(Rot,N)
```

$$integranden := u \quad \textbf{(4.3.6)}$$

```
> Int(Int(integranden,u=0..1),v=0..2)=int(int(integranden,u=0..1),v=0..2)
```

$$\int_0^2 \left(\int_0^1 u\, \mathrm{d}u \right) \mathrm{d}v = 1 \quad \textbf{(4.3.7)}$$